




HOW TO SET UP A KEYPAD ON AN ARDUINO

Posted by Krishna Pattabiraman | Arduino | 26



Keypads are a great way to let users interact with your project. You can use them to navigate menus, enter passwords, and control games and robots.

In this tutorial, I'll show you how to setup a keypad on the [Arduino](#). First I'll explain how the Arduino detects key presses, then I'll show you how to find the pinout of any keypad. As a simple example, I'll show you how to print out the key presses on the serial monitor and an LCD. Finally, I'll show you how to activate a 5V relay when a password is entered correctly.

BONUS: I made a quick start guide for this tutorial that you can [download](#) and go back to later if you can't set this up right now. It covers all of the steps, diagrams, and code you need to get started.

I'll be using a [4X4 matrix membrane keypad](#) in this article, but there's also code and wiring diagrams for [3X4 matrix keypads](#) as well. I like membrane style keypads because they're thin and they also have adhesive backing so you can stick them to most flat surfaces. You can also get [telephone style keypads](#) that have thicker buttons if you like that style better. Even salvaged keypads from old telephones will work with the Arduino.




HOW KEYPADS WORK

The buttons on a keypad are arranged in rows and columns. A 3X4 keypad has 4 rows and 3 columns, and a 4X4 keypad has 4 rows and 4 columns:




Beneath each key is a membrane switch. Each switch in a row is connected to the other switches in the row by a conductive trace underneath the pad. Each switch in a column is connected the same way – one side of the switch is connected to all of the other switches in that column by a conductive trace. Each row and column is brought out to a single pin, for a total of 8 pins on a 4X4 keypad:



Pressing a button closes the switch between a column and a row trace, allowing current to flow between a column pin and a row pin.


The schematic for a 4X4 keypad shows how the rows and columns are connected:




The Arduino detects which button is pressed by detecting the row and column pin that's connected to the button.

This happens in four steps:


1. First, when no buttons are pressed, all of the column pins are held HIGH, and all of the row pins are held LOW:




2. When a button is pressed, the column pin is pulled LOW since the current from the HIGH column flows to the LOW row pin:



3. The Arduino now knows which column the button is in, so now it just needs to find the row the button is in. It does this by switching each one of the row pins HIGH, and at the same time reading all of the column pins to detect which column pin returns to HIGH:




4. When the column pin goes HIGH again, the Arduino has found the row pin that is connected to the button:




From the diagram above, you can see that the combination of row 2 and column 2 could only mean that the number 5 button was pressed.

CONNECT THE KEYPAD TO THE ARDUINO

The pin layout for most membrane keypads will look like this:




Follow the diagrams below to connect the keypad to an Arduino Uno, depending on whether you have a 3X4 or 4X4 keypad:



HOW TO FIND THE PINOUT OF YOUR KEYPAD

If your keypad's pin layout doesn't match the ones above, you can probe the pins to figure it out. You'll need to build a test circuit by connecting an LED and a current limiting resistor to the Arduino (or any 5V power source) like this:



First, find out which keypad pins are connected to the button rows. Insert the ground (black) wire into the first pin on the left. Press any button in row 1 and hold it down. Now insert the positive (red) wire into each one of the other pins. If the LED lights up at one of the pins, press and hold another button in row 1, then insert the positive wire into each one of the other pins again. If the LED lights up on a different pin, it means the ground wire is inserted into the row 1 pin. If none of the buttons in row 1 make the LED light up, the ground wire is not connected to row 1. Now move the ground wire over to the next pin, press a button in a different row, and repeat the process above until you've found the pin for each row.

To figure out which pins the columns are connected to, insert the ground wire into the pin you know is row 1. Now press and hold any one of the buttons in that row. Now insert the positive wire into each one of the remaining pins. The pin that makes the LED light up is the pin that's connected to that button's column. Now press down another button in the same row, and insert the positive wire into each one of the other pins. Repeat this process for each one of the other columns until you have each one mapped out.



PROGRAMMING THE KEYPAD

For a basic demonstration of how to setup the keypad, I'll show you how to print each key press to the serial monitor.

INSTALL THE LIBRARY

We'll use the [Keypad library](#) by Mark Stanley and Alexander Brevig. This library takes care of setting up the pins and polling the different columns and rows. To install the Keypad library, go to Sketch > Include Library > Manage Libraries and search for "keypad". Click on the library, then click install.

THE CODE FOR A 4X4 KEYPAD

Once the Keypad library is installed, you can upload this code to the Arduino if you're using a 4X4 keypad:

```
#include <Keypad.h>

const byte ROWS = 4;
const byte COLS = 4;

char hexaKeys[ROWS][COLS] = {
  {'1', '2', '3', 'A'},
  {'4', '5', '6', 'B'},
  {'7', '8', '9', 'C'},
  {'*', '0', '#', 'D'}
};

byte rowPins[ROWS] = {9, 8, 7, 6};
byte colPins[COLS] = {5, 4, 3, 2};

Keypad customKeypad = Keypad(makeKeymap(hexaKeys), rowPins, colPins,
ROWS, COLS);

void setup(){
  Serial.begin(9600);
}

void loop(){
  char customKey = customKeypad.getKey();

  if (customKey){
    Serial.println(customKey);
  }
}
```

THE CODE FOR A 3X4 KEYPAD

If you're using a 3X4 keypad, you can use this code:

```
#include <Keypad.h>

const byte ROWS = 4;
const byte COLS = 3;

char hexaKeys[ROWS][COLS] = {
  {'1', '2', '3'},
  {'4', '5', '6'},
  {'7', '8', '9'},
  {'*', '0', '#'}
};

byte rowPins[ROWS] = {9, 8, 7, 6};
byte colPins[COLS] = {5, 4, 3};

Keypad customKeypad = Keypad(makeKeymap(hexaKeys), rowPins, colPins,
ROWS, COLS);

void setup(){
  Serial.begin(9600);
}
```



```
void loop(){
  char customKey = customKeypad.getKey();


  if (customKey){
    Serial.println(customKey);
  }
}
```

Lines 3 and 4 in the code above set the number of rows and columns on the keypad.

Lines 6-11 define which characters are printed when a particular button is pressed on the keypad. The characters are laid out just as they appear on the keypad. If your keypad has a different layout, you can define which characters are printed when you press a button. For example, say your keypad has a column of letters on the left instead of the right. You would just change it to this:

```
char hexaKeys[ROWS][COLS] = {
  {'A', '1', '2', '3'},
  {'B', '4', '5', '6'},
  {'C', '7', '8', '9'},
  {'D', '*', '0', '#'}
};
```

After you upload the code, open the serial monitor. When you press a key, the value will be printed out:



USING AN LCD WITH THE KEYPAD

Now let's see how to print the key presses on an LCD. 4X4 keypads use 8 pins and 3X4 keypads use 7 pins. That takes up a lot of pins, so I'm going to use an I2C enabled LCD because it only needs 4 wires to connect to the Arduino.

INSTALL THE LIQUIDCRYSTAL_I2C LIBRARY

To use an I2C enabled LCD on the Arduino, you'll need to install the [LiquidCrystal I2C library](#) by Marco Schwartz. This library is nice because it includes most of the functions available in the standard LiquidCrystal library. To install it, download the ZIP file below, then go to Sketch > Include Library > Add .ZIP Library:



[LiquidCrystal I2C.zip](#)

THE WIRE LIBRARY


The Wire library is needed to add support for I2C communication. It comes packaged with the Arduino IDE, so there's no need to install it. But if for some reason it's not installed on your system, go to Sketch > Include Library > Manage Libraries and search for "wire" to install it.

CONNECT THE KEYPAD AND LCD

Once the libraries are installed, connect the ground and Vcc pins of the LCD to the Arduino, then connect the LCD's SDA and SCL pins according to the table below for the different Arduino boards:

Arduino	SDA Pin	SCL Pin
Uno	A4	A5
Nano	A4	A5
Mini	A4	A5
101	SDA	SCL
Zero	SDA	SCL
Leonardo	2	3
Micro	2	3
Due	20	21
Mega	20	21

Then connect the keypad to the Arduino. It should look something like this (for an Arduino Uno):



CODE FOR OUTPUT TO AN LCD

Once everything is connected, upload this code to the Arduino:

```
#include <Wire.h>
#include <LiquidCrystal_I2C.h>
#include <Keypad.h>

const byte ROWS = 4;
const byte COLS = 4;

char hexaKeys[ROWS][COLS] = {
  {'1', '2', '3', 'A'},
  {'4', '5', '6', 'B'},
  {'7', '8', '9', 'C'},
  {'*', '0', '#', 'D'}
```



```
    {'*', '0', '#', 'D'}
};

byte rowPins[ROWS] = {9, 8, 7, 6};
byte colPins[COLS] = {5, 4, 3, 2};

Keypad customKeypad = Keypad(makeKeymap(hexaKeys), rowPins, colPins,
ROWS, COLS);

LiquidCrystal_I2C lcd(0x21, 16, 2);

void setup(){
  lcd.backlight();
  lcd.init();
}

void loop(){
  char customKey = customKeypad.getKey();
  if (customKey){
    lcd.clear();
    lcd.setCursor(0, 0);
    lcd.print(customKey);
  }
}
```

You'll need to add the I2C address of your LCD on line 20:

```
LiquidCrystal_I2C lcd(0x21, 16, 2);
```

The I2C address of my LCD is 0x21, but your's will probably be different. The I2C address of your LCD should be provided in the datasheet, but if not, you can find it by running this [I2C_Scanner sketch](#).

USE A PASSWORD TO ACTIVATE A RELAY

One of the most useful applications of a keypad is to use it for keyed entry. You can set up a password and have the Arduino activate a relay or some other module if the password is correct. The following code will activate a 5V relay when the password is entered correctly:

```
#include <Wire.h>
#include <LiquidCrystal_I2C.h>
#include <Keypad.h>

#define Password_Length 8

int signalPin = 12;

char Data[Password_Length];
char Master[Password_Length] = "123A456";
byte data_count = 0, master_count = 0;
bool Pass_is_good;
char customKey;

const byte ROWS = 4;
const byte COLS = 4;

char hexaKeys[ROWS][COLS] = {
  {'1', '2', '3', 'A'},
  {'4', '5', '6', 'B'},
  {'7', '8', '9', 'C'},
  {'*', '0', '#', 'D'}
};

byte rowPins[ROWS] = {9, 8, 7, 6};
byte colPins[COLS] = {5, 4, 3, 2};

Keypad customKeypad = Keypad(makeKeymap(hexaKeys), rowPins, colPins,
ROWS, COLS);

LiquidCrystal_I2C lcd(0x21, 16, 2);

void setup(){
  lcd.init();
```

```
    lcd.backlight();
    pinMode(signalPin, OUTPUT);
}

void loop(){

    lcd.setCursor(0,0);
    lcd.print("Enter Password:");

    customKey = customKeypad.getKey();
    if (customKey){
        Data[data_count] = customKey;
        lcd.setCursor(data_count,1);
        lcd.print(Data[data_count]);
        data_count++;
    }

    if(data_count == Password_Length-1){
        lcd.clear();

        if(!strcmp(Data, Master)){
            lcd.print("Correct");
            digitalWrite(signalPin, HIGH);
            delay(5000);
            digitalWrite(signalPin, LOW);
        }
        else{
            lcd.print("Incorrect");
            delay(1000);
        }

        lcd.clear();
        clearData();
    }
}

void clearData(){
    while(data_count !=0){
        Data[data_count--] = 0;
    }
    return;
}
```

You can change the password on line 10 by replacing the `123A456` text with your own password:

```
char Master[Password_Length] = "123A456";
```

The length of the password needs to be set on line 5:

```
#define Password_Length 8
```


The password in the example above is only 7 characters long, but the password length is actually one greater than 7 because there is a null character added to the end of the string. For example, if your password is 5 characters long, you would enter 6 for the password length.

The output pin that activates the relay is defined on line 7:

```
int signalPin = 12;
```

After connecting everything to the Arduino, you should have something that looks like this:





Well that's about it. It's not hard to set up a keypad at all. I think with a little trial and error you should be able to modify the code above to work with most of the projects you'd want to use a keypad for. But if you run into problems, just let us know in the comments and we'll try to help you out.

Krishna Pattabiraman is a frequent guest writer on Circuit Basics and the founder of www.codeelectron.com.

